

Scientific Computing TU Berlin Winter 2021/22 © Jürgen Fuhrmann Notebook 04

- `begin`
- `using PlutoUI`
- `using LinearAlgebra`
- `using InteractiveUtils`
- `import AbstractTrees`
- `end`

Julia type system

Concrete types

Structs

Functions, Methods and Multiple Dispatch

Abstract types

 The type tree

The power of multiple dispatch

- `TableOfContents(title="")`

Julia type system

- Julia is a strongly typed language
- Knowledge about the layout of a value in memory is encoded in its type
- Prerequisite for performance
- There are concrete types and abstract types
- See the [Julia WikiBook](#) for more

Concrete types

- Every value in Julia has a concrete type
- Concrete types correspond to computer representations of objects
- Inquire type info using `typeof()`

Built-in types

- Default types are deduced from concrete representations

Int64

- `typeof(10)`

Float64

- `typeof(10.0)`

ComplexF64 (alias for Complex{Float64})

- `typeof(3.0+3im)`

Irrational{ π }

- `typeof(π)`

Bool

- `typeof(false)`

String

- `typeof("false")`

Vector{Float16} (alias for Array{Float16, 1})

- `typeof(Float16[1,2,3])`

Matrix{Int64} (alias for Array{Int64, 2})

- `typeof(rand(Int,3,3))`

- One can initialize a variable with an explicitly given fixed type.

10

- `i::Int8=10`

Int8

- `typeof(i)`

5.0

- `x::Float16=5.0`

Float16

- `typeof(x)`

Structs

- Structs allow to define custom types

```

• struct MyColor64
•     r::Float64
•     g::Float64
•     b::Float64
• end

```

```
c = MyColor64(0.1, 0.2, 0.3)
```

```
• c=MyColor64(0.1,0.2,0.3)
```

```
MyColor64
```

```
• typeof(c)
```

```
24
```

```
• sizeof(c)
```

```
0.1
```

```
• c.r
```

setfield! immutable struct of type MyColor64 cannot be changed

1. `setproperty!` @ Base.jl:34 [inlined]
2. `top-level scope` @ **Local: 1** [inlined]

```
• c.r=10
```

- Structs can be parametrized with types. This is similar to array types which are parametrized by their element types

```

• struct MyColor{T}
•     r::T
•     g::T
•     b::T
• end

```

```
c2 = MyColor(4, 25, 233)
```

```
• c2=MyColor(4,25,233)
```

```
MyColor{Int64}
```

```
• typeof(c2)
```

```
24
```

```
• sizeof(c2)
```

```
c3 = MyColor(0x04, 0x19, 0xe9)
```

- `c3=MyColor{UInt8}(4,25,233)`

```
MyColor{UInt8}
```

- `typeof(c3)`

```
3
```

- `sizeof(c3)`

Functions, Methods and Multiple Dispatch

- Functions can have different variants of their implementation depending on the types of parameters passed to them
- These variants are called **methods**
- All methods of a function `f` can be listed calling `methods(f)`
- The act of figuring out which method of a function to call depending on the type of parameters is called **multiple dispatch**

- `test_dispatch(x)="general case: $(typeof(x)), x=$(x)";`

- `test_dispatch(x::Float64)="special case Float64, x=$(x)";`

- `test_dispatch(x::Int64)="special case Int64, x=$(x)";`

```
"special case Int64, x=3"
```

- `test_dispatch(3)`

```
"general case: Bool, x=false"
```

- `test_dispatch(false)`

```
"special case Float64, x=3.0"
```

- `test_dispatch(3.0)`

Here we defined a generic method which works for any variable passed. In the case of `Int64` or `Float64` parameters, special cases are handled by different methods of the same function. The compiler decides which method to call. This approach allows to specialize implementations dependent on data types, e.g. in order to optimize performance.

The `methods` function can be used to figure out which methods of a function exists.

3 methods for generic function `test_dispatch`:

- `test_dispatch(x::Float64)` in `Main.workspace2` at [/home/fuhrmann/Wias/teach/scicomp/pluto/nb04-julia-types.jl#=#0468c2da-0955-11eb-271b-5d84d5d8343d:1](#)
- `test_dispatch(x::Int64)` in `Main.workspace2` at [/home/fuhrmann/Wias/teach/scicomp/pluto/nb04-julia-types.jl#=#0cc7808a-0955-11eb-0b4d-ff491af88cf5:1](#)
- `test_dispatch(x)` in `Main.workspace2` at [/home/fuhrmann/Wias/teach/scicomp/pluto/nb04-julia-types.jl#=#f5cc25e6-0954-11eb-179b-eddff99dd392:1](#)

• `methods(test_dispatch)`

The function/method concept somehow corresponds to **C++14 generic lambdas**

```
auto myfunc=[](auto &y, auto &y)
{
    y=sin(x);
};
```

is equivalent to

```
function myfunc!(y,x)
    y=sin(x)
end
```

Many **generic programming** approaches possible in C++ also work in Julia,

If not specified otherwise via parameter types, Julia functions are generic: "automatic auto"

Abstract types

- Abstract types label concepts which work for a several concrete types without regard to their memory layout etc.
- All variables with concrete types corresponding to a given abstract type (should) share a common interface
- A common interface consists of a set of functions with methods working for all types exhibiting this interface
- The functionality of an abstract type is implicetly characterized by the methods working on it
- This concept is close to **"duck typing"**: use the "duck test" — "If it walks like a duck and it quacks like a duck, then it must be a duck" — to determine if an object can be used for a particular purpose
- When trying to force a parameter to have an abstract type, it ends up with having a concrete type which is compatible with that abstract type

The type tree

- Types can have subtypes and a supertype
- Concrete types are the leaves of the resulting type tree
- Supertypes are necessarily abstract
- There is only one supertype for every (abstract or concrete) type
- Abstract types can have several subtypes

[BigFloat, Float16, Float32, Float64]

- `subtypes(AbstractFloat)`

- Concrete types have no subtypes

[]

- `subtypes(Float64)`

AbstractFloat

- `supertype(Float64)`

[BigFloat, Float16, Float32, Float64]

- `subtypes(AbstractFloat)`

Real

- `supertype(AbstractFloat)`

Number

- `supertype(Real)`

Any

- `supertype(Number)`


- "Any" is the root of the type tree and has itself as supertype

Any

- `supertype(Any)`

We can use the `AbstractTrees` package to visualize the type tree. We just need to define what it means to have children for a type.

```
• AbstractTrees.children(x::Type) = subtypes(x)
```



```
• AbstractTrees.Tree(Number)
```

There are operators for testing type relationships

```
true
```

```
• Float64<: Number
```

```
false
```

```
• Float64<: Integer
```

```
false
```

```
• isa(3,Float64)
```

```
true
```

```
• isa(3.0,Float64)
```

Abstract types can be used for method dispatch as well

```
dispatch2 (generic function with 2 methods)
```

```
• begin
•   dispatch2(x::AbstractFloat)="$(typeof(x)) <:AbstractFloat, x=$(x)"
•   dispatch2(x::Integer)="$(typeof(x)) <:Integer, x=$(x)"
• end
```

```
"Int64 <:Integer, x=13"
```

```
• dispatch2(13)
```

```
"Float16 <:AbstractFloat, x=3.0"
```

- `dispatch2(Float16(3))`

The power of multiple dispatch

- Multiple dispatch is one of the defining features of Julia
- Combined with the hierarchical type system it allows for powerful generic program design
- New datatypes (different kinds of numbers, differently stored arrays/matrices) work with existing code once they implement the same interface as existent ones.
- In some respects C++ comes close to it, but for the price of more and less obvious code